
Programación declarativa (26–Enero–2005)

Apellidos:

Nombre:

Notas

1. En la evaluacíon de los ejercicios se tendrá en cuenta la simplicidad y la eficiencia.

2. Hay que describir las definiciones auxiliares.

Ejercicio 1 [7 puntos] Definir la relacíon recorrido_completo(+N,-R) que se verifique si
R es un recorrido completo de un caballo de ajedrez en un tablero de dimensiones NxN desde la
posicíon 1/1 (es decir, es un recorrido del caballo que comienza en la casilla 1/1, pasa por todas las
casillas del tablero pero sin pasar dos veces por la misma casilla). Por ejemplo,

?- recorrido_completo(5,R).
R = [1/1, 3/2, 5/3, 4/5, 2/4, 1/2, 3/3, 4/1, 2/2, 1/4,

3/5, 5/4, 4/2, 2/1, 1/3, 2/5, 4/4, 5/2, 3/1, 2/3,
1/5, 3/4, 5/5, 4/3, 5/1]

?- recorrido_completo(4,R).
No

Solución: La definicíon derecorrido_completo es

recorrido_completo(N,R) :-
M is N*N,
length(R1,M),
recorrido(N,_,1/1,R1),
reverse(R1,R).

La relacíon recorrido(+N,+P1,+P2,-R) se verifica siR es un recorrido de un caballo
en un tablero de dimensionesNxNdesde la posición P1 a laP2 sin pasar dos veces por la misma
casilla. Por ejemplo,

?- recorrido(4,1/1,3/2,R).
R = [1/1, 2/3, 4/2, 3/4, 1/3, 3/2]

La definicíon derecorrido es

recorrido(N,P1,P2,R) :-
recorrido_aux(N,P1,[P2],R).

La relacíon recorrido_aux(+N,+P1,+RP,-R) se verifica siR es un recorrido de un
caballo en un tablero de dimensionesNxN compuesto de un recorrido desdeP1 hasta el primer
elemento del recorrido parcialRP(con nodos distintos a los deRP) junto RP.

recorrido_aux(_,P,[P|R],[P|R]).
recorrido_aux(N,P1,[P2|R1],R) :-

salto(N,P2,P3),
not(member(P3,[P2|R1])),
recorrido_aux(N,P1,[P3,P2|R1],R).



La relacíonsalto(+N,+P1,?P2) se verifica siP2 es una posición a donde puede moverse
un caballo desde la posiciónP1 en un tablero de dimensionesN*N. Por ejemplo,

?- salto(5,1/4,P).
P = 3/5 ;
P = 3/3 ;
P = 2/2 ;
No

La ddefinicíon desalto es

salto(N,X1/Y1,X2/Y2) :-
dxy(Dx,Dy),
X2 is X1+Dx,
1 =< X2, X2 =< N,
Y2 is Y1+Dy,
1 =< Y2, Y2 =< N.

La relacíondxy(?X,?Y) se verifica si un caballo puede moverseX espacios horizontales eY
verticales.

dxy(2,1).
dxy(2,-1).
dxy(-2,1).
dxy(-2,-1).
dxy(1,2).
dxy(1,-2).
dxy(-1,2).
dxy(-1,-2).

Ejercicio 2 [3 puntos] Se considera la siguiente gramática de cĺausulas definidas

s --> n0.
s --> n1, s, n2.
n0 --> [a].
n1 --> [p].
n2 --> [q].

1. Traducir las reglas gramaticales a reglas Prolog.

2. Explicar, mediante eĺarbol de derivacíon, ćomo se obtienen las 2 primeras respuestas a la
pregunta

?- s(X,[]).

3. Escribir las 5 primeras respuestas a la pregunta

?- s(X,[]).

Solución:

1. El programa Prolog correspondiente a la gramática es



s(A, B) :- n0(A, B).
s(A, B) :- n1(A, C), s(C, D), n2(D, B).
n0([a|A], A).
n1([p|A], A).
n2([q|A], A).

3. Las cinco primeras respuestas son

X = [a]
X = [p, a, q]
X = [p, p, a, q, q]
X = [p, p, p, a, q, q, q]
X = [p, p, p, p, a, q, q, q, q]


